

COUNTING ALL EQUILATERAL TRIANGLES IN $\{0, 1, \dots, n\}^3$

E. J. IONASCU

ABSTRACT. We describe a procedure of counting all equilateral triangles in the three dimensional space whose coordinates are allowed only in the set $\{0, 1, \dots, n\}$. This sequence is denoted here by $ET(n)$ and it has the entry A102698 in “The On-Line Encyclopedia of Integer Sequences”. The procedure is implemented in Maple and its main idea is based on the results in [3]. Using this we calculated the values $ET(n)$ for $n = 1 \dots 55$ extending previous calculations known for $n \leq 34$. Some facts and conjectures about this sequence are stated. The main conjecture raised here is that $\lim_{n \rightarrow \infty} \frac{\ln ET(n)}{\ln n + 1}$ exists.

1. INTRODUCTION

If we restrict the vertices of an equilateral triangle to be in \mathbb{Z}^3 we obtain a typical element in $\mathcal{ET}(\mathbb{Z})$. It is not that hard to see that there are no such triangles whose vertices are contained in the coordinate planes or any other plane parallel to one of them. Also, the sides of a triangle in $\mathcal{ET}(\mathbb{Z})$ cannot be of an arbitrary length. If one such triangle is considered, a whole family in $\mathcal{ET}(\mathbb{Z})$ can be generated from it with vertices in the same plane. Moreover, we have shown in [3] the following theorems that we are going to use in our construction here.

Theorem 1.1. *If the triangle $\triangle OPQ \in \mathcal{ET}(\mathbb{Z})$ with O the origin and $l = \|\vec{OP}\|$ then:*

- (i) *the points P and Q are contained in a plane of the equation $ax + by + cz = 0$, where*

$$(1) \quad a^2 + b^2 + c^2 = 3d^2, \quad a, b, c, d \in \mathbb{Z}$$

and $l^2 = 2q$;

- (ii) *the side length l is of the form $\sqrt{2(m^2 - mn + n^2)}$ with $m, n \in \mathbb{Z}$.*

It is important to be able to generate all primitive solutions of (1):

Received January 3, 2007; revised September 21, 2007.
2000 *Mathematics Subject Classification.* Primary 11B99; Secondary 11D09, 11C08.
Key words and phrases. diophantine equations; integers.

Theorem 1.2. *The following formulae give a three integer parameter solution of (1):*

$$(2) \quad \begin{cases} a = -x_1^2 + x_2^2 + x_3^2 - 2x_1x_2 - 2x_1x_3 \\ b = x_1^2 - x_2^2 + x_3^2 - 2x_2x_1 - 2x_2x_3 \\ c = x_1^2 + x_2^2 - x_3^2 - 2x_3x_1 - 2x_3x_2 \\ d = x_1^2 + x_2^2 + x_3^2, \end{cases} \quad x_1, x_2, x_3 \in \mathbb{Z}.$$

Moreover, every solutions of (1), a, b, c, d can be found from (2) with $x_1, x_2, x_3 \in \mathbb{Z}[\frac{1}{\sqrt{k}}]$ and $k = (3d - a - b - c)/2 \in \mathbb{N}$.

We include some general remarks about the solutions of (1) which are discussed in [3]:

- if we assume that $\gcd(a, b, c) = 1$ then all a, b, c, d must be odd integers
- for every d odd there exists at least one solution which is not trivial (i.e. $a = b = c = d$)
- [Gauss] a positive integer n can be written as a sum of three squares iff n is not of the form $4^k(8l + 7)$ with $k, l \in \mathbb{Z}$ (see [1] for an elementary proof)

Our construction depends essentially on a particular solution, $(r, s) \in \mathbb{Z}^2$, of the equation:

$$(3) \quad 2(a^2 + b^2) = s^2 + 3r^2.$$

It turns out that this Diophantine equation has always solutions if a, b, c and d are integers satisfying (1). The family of equilateral triangles we have mentioned can be described more precisely as another parametrization.

Theorem 1.3. *Let a, b, c, d be odd positive integers such that $a^2 + b^2 + c^2 = 3d^2$, with $\gcd(d, c) = 1$. Then a triangle $\triangle OPQ \in \mathcal{ET}(\mathbb{Z})$ has its vertices in the plane of the equation $a\alpha + b\beta + c\gamma = 0$ if and only if $P(u, v, w)$ and $Q(x, y, z)$ are given by*

$$(4) \quad \begin{cases} u = m_u m - n_u n, \\ v = m_v m - n_v n, \\ w = m_w m - n_w n, \end{cases} \quad \text{and} \quad \begin{cases} x = m_x m - n_x n, \\ y = m_y m - n_y n, \\ z = m_z m - n_z n, \end{cases}$$

with

$$\begin{cases} m_x = -\frac{1}{2}[db(3r + s) + ac(r - s)]/q, & n_x = -(rac + dbs)/q \\ m_y = \frac{1}{2}[da(3r + s) - bc(r - s)]/q, & n_y = (das - bcr)/q \\ m_z = (r - s)/2, & n_z = r \end{cases}$$

(5) and

$$\begin{cases} m_u = -(rac + dbs)/q, & n_u = -\frac{1}{2}[db(s - 3r) + ac(r + s)]/q \\ m_v = (das - rbc)/q, & n_v = \frac{1}{2}[da(s - 3r) - bc(r + s)]/q \\ m_w = r, & n_w = (r + s)/2 \end{cases}$$

where $q = a^2 + b^2$, (r, s) is a suitable solution of (3) and $m, n \in \mathbb{Z}$.

Moreover, the side-lengths of such a triangle are equal to $d\sqrt{2(m^2 - mn + n^2)}$.

Let us observe that one can use Theorem 1.3 as long as $\min(\gcd(q, c), \gcd(q, a), \gcd(d, b)) = 1$. However, we have calculated that this property holds true for all solutions a, b, c, d of (1) such that $\gcd(a, b, c) = 1$ and with all odd $d < 1105$. In fact, 1105 is the first d for which the condition above fails with certain choices of a, b, c . This allows us to calculate $ET(n)$ for $n = 1, \dots, 55$ as we will see later.

2. DESCRIPTION OF THE PROCEDURE AND SOME INGREDIENTS

The idea is based on the facts above and a few other results. One would like first to find the side lengths of the triangles in $\mathcal{ET}(\mathbb{Z}) \cap \mathcal{C}_n$, where \mathcal{C}_n is the cube $[0, n]^3$. This will partition these triangles into clear classes. For this purpose we will use the Proposition 1.2. Then for a given side-length l we need to find all the possible planes that contain triangles of sides l . This gives another criteria of sub-partition even further these triangles. Using the parametrizations given in Theorem 1.3 then one finds such smallest triangle within a given plane that can fit in \mathcal{C}_n after a translation and rotation. Once that is obtained we rotate and translate it remaining in \mathcal{C}_n in all possible ways, but in a pairwise disjoint manner. A formula for the number of all these equilateral triangles will be given. Finally all these numbers are added up to obtain $ET(n)$.

The first fact that we will use is the following geometric observation.

Proposition 2.1. *The largest side length of an equilateral triangle whose vertices are contained in a cube of side lengths r is $r\sqrt{2}$.*

Proof. If an equilateral triangle of side length l , has its vertices in the cube $[0, r]^3$, we denote by A_1, A_2 , and A_3 the areas of the three projections of this triangle on the three coordinate planes. It is easy to see that $A^2 = A_1^2 + A_2^2 + A_3^2$ where A is the area of the given equilateral triangle. It is easy to see that the maximum of the area of an arbitrary triangle inscribed in a square of side length r is $r^2/2$. Hence $A^2 = 3l^4/16 \leq 3r^4/4$. This gives $l \leq r\sqrt{2}$. Certainly this maximum is attained when the vertices of the triangle are at the corners of the cube such that every two are diagonally opposite on the face they belong. \square

Let us work out a concrete example ($n = 4$) to exemplify our counting. Using Proposition 2.1 and the part (ii) of Theorem 1.1, the side lengths of the triangles in $\mathcal{ET}(\mathbb{Z}) \cap \mathcal{C}_4$ can only be $\sqrt{2}, \sqrt{6}, 2\sqrt{2}, \sqrt{14}, 3\sqrt{2}, 2\sqrt{6}, \sqrt{26}$ or $4\sqrt{2}$. The d values here are 1 or 3. Since $3(1^2) = 1^2 + 1^2 + 1^2$ and $3(3^2) = 1^2 + 1^2 + 5^2$ are the only solutions of (1) for $d = 1$ or $d = 3$, the parametrizations we need in this case are, as shown in [3]:

$$\mathcal{T}_{1,1,1} = \{[(0, 0, 0), (m, -n, n - m), (m - n, -m, n)] : m, n \in \mathbb{Z}, m^2 + n^2 \neq 0\}$$

and

$$\mathcal{T}_{1,1,5} = \{[(0, 0, 0), (4m - 3n, m + 3n, -m), (3m + n, -3m + 4n, -n)] : m, n \in \mathbb{Z}, m^2 + n^2 \neq 0\}.$$

Here we used the notation $\mathcal{T}_{a,b,c}$ for all triangles in $\mathcal{ET}(\mathbb{Z})$ having the origin as one its vertices and the other two contained in the plane $\{(\alpha, \beta, \gamma) : a\alpha + b\beta + c\gamma = 0\}$.

Using the first parametrization we find the m, n such that the triangle obtained after a translation fits in \mathcal{C}_4 and the side lengths are $\sqrt{2}$: $T_1 = \{(1, 0, 0), (0, 1, 0), (0, 0, 1)\}$. This triangle can be translated in various ways inside of \mathcal{C}_4 , and together with all its cube symmetries and their translations contribute with a total of 512 in $ET(4)$. We will prove formula (6) that gives in particular the total of all different triangles generated by T_1 inside of \mathcal{C}_n under translations and rotations. The first parametrization needs to be used also for all the side lengths $\sqrt{6}, 2\sqrt{2}, \sqrt{14}, 3\sqrt{2}, 2\sqrt{6}, \sqrt{26}$ and $4\sqrt{2}$. The corresponding triangles obtained respectively are: $T_2 := \{(1, 0, 2), (2, 1, 0), (0, 2, 1)\}$, $2T_1, T_3 := \{(2, 0, 3), (0, 3, 2), (3, 2, 0)\}$, $3T_1, 2T_2, T_4 := \{(1, 4, 0), (4, 0, 1), (0, 1, 4)\}$ and $4T_1$. Using formula (6) one can check that the contribution of each of these triangles to $ET(4)$ is (respectively): 216, 216, 128, 64, 8, 16 and 8.

There is a need to use the second parametrization since one can take $d = 3$ to obtain the side length $3\sqrt{2}$. This gives still a new triangle: $T_5 := \{(0, 0, 1), (1, 4, 0), (4, 1, 0)\}$. This triangle and all the ones generated inside of \mathcal{C}_4 by its transformations described before are different of the ones above because they are contained in planes having different normals. The contribution of T_5 to $ET(4)$ is 96. If tallying all these classes one gets that $ET(4) = 1264$.

As we can see from this example, one has to derive a way of finding how many other triangles can generate from a given one, say T , inside of \mathcal{C}_n under all possible translations, cube symmetries and their translations. We are going to call these transformations of a triangle *allowed* transformations since we have to make this a little more precise. We are going to assume that the given triangle T that is inside \mathcal{C}_t is minimal in the sense that at least one of the coordinates of the vertices in T is zero and t is the smallest dimension k of the cube \mathcal{C}_k containing T or one of its images given by an allowed transformations on it.

Let us denote by $O(T)$ the orbit generated by T within \mathcal{C}_t under all allowed transformations. We also need to introduce the standard unit vectors $e_1 = (1, 0, 0)$, $e_2 = (0, 1, 0)$ and $e_3 = (0, 0, 1)$.

It is actually surprising that in order to compute the number f , of all distinct triangles generated by T (union of all translations of $O(T)$) within \mathcal{C}_n ($n \geq t$) one just needs the following five variables that depend only on T :

- (i) n – the dimension of the cube,
- (ii) t – the maximum of all the coordinates in T ,
- (iii) $\alpha(T)$ – the cardinality of $O(T)$,
- (iv) $\beta(T)$ – the cardinality of $O(T) \cap [O(T) + e_1]$,
- (v) $\gamma(T)$ – the cardinality of $[O(T) + e_1] \cap [O(T) + e_2]$.

Theorem 2.1. *The function $f(T, n)$ described above is given by*

$$(6) \quad f(T, n) = (n + 1 - t)^3 \alpha - 3(n + 1 - t)^2 (n - t) \beta + 3(n + 1 - t)(n - t)^2 \gamma,$$

for all $n \geq t$.

Proof. Let us consider the cube $\mathcal{C}_s = \{0, 1, \dots, s\}^3$ where $s = n - t$. Clearly the number of points in this set is $(s + 1)^3$. Each point p in the set \mathcal{C}_s is considered here as a vector. So, $f = |\bigcup_{p \in \mathcal{C}_s} O(T) + p|$. One essential observation here is such that $|[O(T) + p] \cap [O(T) + q]| = 0$ for every p, q such that $\|p - q\| \geq 1$, where $\|p - q\| = \min_{i=1,2,3} (|p_i - q_i|)$. This is due to the minimality of T .

Let us write the elements of \mathcal{C}_s in lexicographical order: p_1, p_2, \dots, p_k where $k = (s + 1)^3$. We look now at \mathcal{C}_s as the three dimensional grid graph. Faces in this graph are simply unit squares formed by vertices from \mathcal{C}_s . One can look at the cardinality of $\bigcup_{i=1..j} O(T) + p_i$ and show by induction on j that this is equal to

$$j\alpha - (\#\text{edges}(\mathcal{C}_s(j))\beta + (\#\text{faces}(\mathcal{C}_s(j))\gamma$$

where $\mathcal{C}_s(j)$ is the graph induced in \mathcal{C}_s by the vertices p_1, p_2, \dots, p_j . Hence we just need to compute the number of edges and faces in \mathcal{C}_s . There are eight vertices in this graph that have degree 3 (the corners), there are $(s - 1)^3$ vertices of degree 6, also $6(s - 1)^2$ vertices with degree 5 and finally $12(s - 1)$ of degree 4. This gives a total of

$$\frac{1}{2} [24 + 6(s - 1)^3 + 30(s - 1)^2 + 48(s - 1)] = 3s(s + 1)^2$$

edges. The number of faces is equal to $\frac{1}{2} [6s^3 + 6s^2] = 3s^2(s + 1)$. □

Example. Suppose we take $T = T_5$. Then clearly $t = 4$. One can use a symbolic calculator to find $\alpha(T) = 96, \beta(T) = 24$ and $\gamma(T) = 0$. So, the contribution of T_5 to an arbitrary cube \mathcal{C}_n is $f(T_5, n) = 96(n - 3)^3 - 72(n - 3)^2(n - 4) = 24n(n - 3)^2$.

Remark. These facts give a way to find lower bounds for $ET(n)$. For instance, if we put the contribution of T_1 and T_2 together we obtain that $ET(n) \geq 8(2n - 1)(n^2 - n + 1)$ for all $n \geq 2$.

To generate the side lengths we would like to use a well-known result due to Euler (see [4, pp. 568] and [2, pp. 56]).

Proposition 2.3 (Euler's $6k + 1$). *An integer t can be written as $m^2 - mn + n^2$ for some $m, n \in \mathbb{Z}$ if and only if in the prime factorization of t , 2 and the primes of the form $6k - 1$ appear to an even exponent.*

3. THE CODE

Using Proposition 2.3 and Proposition 2.1 we have the following procedure in Maple to compute the side lengths modulo a factor of two and the square root:

```

> sides:=proc(n)
  local i,j,k,L,a,m,p,q,r,ms;
> L:={1}; ms:=n^2;
> for i from 2 to ms do
  a:=ifactors(i); k:=nops(a[2]); r:=0;
> for j from 1 to k do m:=a[2][j][1]; p:=m mod 6;
  q:=a[2][j][2] mod 2; if r=0 and (m=2 or p=5) and q=1
  then r:=1 fi;

```

```
> if r=0 then L:=L union {i}; fi;
> od; L:=convert(L,list); end:
```

This procedure gives for $n = 10$: [1, 3, 4, 7, 9, 12, 13, 16, 19, 21, 25, 27, 28, 31, 36, 37, 39, 43, 48, 49, 52, 57, 61, 63, 64, 67, 73, 75, 76, 79, 81, 84, 91, 93, 97, 100]. This gives the corresponding side-lengths

```
[√2, √6, 2√2, √14, 3√2, 2√6, √26, 4√2, √38, √42, 5√2, 3√6, 2√14, √62,
6√2, √74, √78, √86, 4√6, 7√2, 2√26, √114, √122, 3√14, 8√2, √134, √146,
5√6, 2√38, √158, 9√2, 2√42, √182, √186, √194, 10√2]
```

We need a procedure that will give the odd values of d that “divide” a certain side length in the sense it is possible to write it as $d\sqrt{m^2 - mn + n^2}$ with $m, n \in \mathbb{Z}$:

```
> dkl:=proc(side)
> local i,x,noft,div,y,y1,z;
> x:=convert(divisors(side),list); noft:=nops(x); div:={ };
> for i from 1 to noft do z:=x[i] mod 2;
> if z=1 then y:=side/x[i]^2;y1:=floor(y); if y=y1
> then div:=div union {x[i]}; fi; fi;
> od;
> convert(div,list); end:
```

For instance, if the $side = \sqrt{882}$ this procedure gives [1, 3, 7, 21] which means we have at least four possible parametrizations that we can use to find minimal equilateral triangles.

Next we need to find all the nontrivial solutions $[a, b, c]$ of (1), given an odd positive integer d , with the property $\gcd(a, b, c) = 1$, $0 < a \leq b \leq c$ which is based on an internal procedure to solve Diophantine equation $A = X^2 + Y^2$:

```
> abcsol:=proc(q) local i,j,k,u,x,y,sol,cd; sol:={};
> for i from 1 to q do
> u:=[isolve(3*d^2-i^2=x^2+y^2)]; k:=nops(u);
> for j from 1 to k do
> if rhs(u[j][1])>=i and rhs(u[j][2])>=i then
> cd:=gcd(gcd(i,rhs(u[j][1])),rhs(u[j][2]));
> if cd=1 then sol:=sol union
> {sort([i,rhs(u[j][1]),rhs(u[j][2]))]}; fi; fi;
> od; od; convert(sol,list); end:
```

For $d = 17$, abcsol finds four different solutions, [[11, 11, 25], [13, 13, 23], [1, 5, 29], [7, 17, 23]], and in a few seconds sends out 333 solutions for $d = 2007$. One interesting solution in this last case is

$$1937^2 + 1973^2 + 2107^2 = 3(2007)^2.$$

Now based on the Theorem 1.3 we take a solution of (1) as given by the procedure above and calculate the general parametrization:

```
> findpar:=proc(a,b,c,m,n)
> local i,j,r,s,sol,mx,nx,my,ny,mu,nu,mv,nv,mz,nz,mw,nw,q,d,u,v,
> w,x,y,z,ef,ns,om,l,t;
> q:=a^2+b^2; sol:=convert({isolve(2*q=x^2+3*y^2)},list);
> ns:=nops(sol); d:=sqrt((a^2+b^2+c^2)/3); ef:=0;
> for i from 1 to ns do
```

```

    if ef=0 then r:=rhs(sol[i][1]); s:=rhs(sol[i][2]);
  > uu:=(s^2+3*r^2-2*q)^2; if uu>0 then t:=s;
  > s:=r; r:=t; fi;
  > mz:=(r-s)/2; nz:=r; mw:=r; nw:=(r+s)/2;
  mx:=-((d*b(3*r+s)+a*c(r-s))/(2*q));
  nx:=-((r*a*c+d*b*s)/q); my:=(d*a(3*r+s)-b*c(r-s))/(2*q);
  ny:=-((r*b*c-d*a*s)/q);
  mu:=-((r*a*c+d*b*s)/q); nu:=-((d*b(s-3r)+a*c(r+s))/(2*q));
  mv:=(d*a*s-r*b*c)/q; nv:=-((d*a(3r-s)+b*c(r+s))/(2*q));
  > if mx=floor(mx) and nx=floor(nx) and my=floor(my) and
  ny=floor(ny) and mu=floor(mu) and nu=floor(nu) and
  mv=floor(mv) and nv=floor(nv) then
  u:=(mu)m-(nu)n; v:=(mv)m-(nv)n; w:=(mw)m-(nw)n;
  x:=(mx)m-(nx)n; y:=(my)m-(ny)n; z:=(mz)m-(nz)n;
  om:=[[u,v,w],[x,y,z]];
  > ef:=1; fi; fi; od; om; end:

```

For the solution, $[1, 5, 29]$, found earlier for the case $d = 17$, *findpar* gives

$$[[-11m - 13n, -21m + 20n, 4m - 3n], [-24m + 11n, -m + 21n, m - 4n]].$$

Next, using this parametrization we would like to find if there is any equilateral triangle in $\mathcal{ET}(\mathbb{Z})$ which after a translation fits inside \mathcal{C}_{stopp} .

```

1: minimaltr:=proc(s,a,b,c,stopp)
2: local i,z,u,nt,d,m,n,T,alpha,beta,gamma,tr,out,L,tri,noft,
  tria,orb,avb,length,lengthn;
3: d:=sqrt((a^2+b^2+c^2)/3); $z:=s/d^2;
  u:=convert({isolve(z=q^2-qr+r^2)},list); nt:=nops(u);
4: for i from 1 to nt do
5: T:=findpar(a,b,c,rhs(u[i][1]),rhs(u[i][2]));
6: alpha:=min(T[1][1],T[2][1],0); beta:=min(T[1][2],T[2][2],0);
  gamma:=min(T[1][3],T[2][3],0);
7: tr[i]:={T[1][1]-alpha,T[1][2]-beta,T[1][3]-gamma,
  T[2][1]-alpha,T[2][2]-beta,T[2][3]-gamma,
  [-alpha,-beta,-gamma]};
8: out[i]:=max(tr[i][1][1],tr[i][1][2],tr[i][1][3],tr[i][2][1],
  tr[i][2][2],tr[i][2][3],tr[i][3][1],tr[i][3][2],tr[i][3][3]);
9: od; L:=sort([seq(out[i],i=1..nt)]); tri:={};
10: for i from 1 to nt do if out[i]<= stopp then
  tri:=tri union {tr[i]}; fi; od;
11: tri:=convert(tri,list); tria:={};
12: if nops(tri)>0 then noft:=nops(tri); tria:={tri[1]};
  orb:=transl(tri[1]);
13: for i from 1 to noft do avb:=evalb(tri[i] in orb);
14: if avb=false then orb:=orb union transl(tri[i]);
  tria:=tria union {tri[i]};
15: fi; od; fi; tria; end:

```

The minimal triangle given by this procedure for $s = 17\sqrt{2}$, $a = 1$, $b = 5$, $c = 29$, $stopp = 30$: $\{[11, 21, 0], [24, 1, 3], [0, 0, 4]\}$. The last part of the procedure is actually searching for a set of triangles that generate all the triangles in $\mathcal{ET}(\mathbb{Z})$ that

lay in planes of normal (a, b, c) or all other 23 possibilities obtained by permuting the coordinates and changing signs. The next procedure is used above and later in order to compute the parameters $\alpha(T)$, $\beta(T)$ and $\gamma(T)$.

```
> transl:=proc(T)
> local S,Q,i,j,k,a2,b2,c2,a,b,c,d;
> Q:=convert(T,list); a:=max(Q[1][1],Q[2][1],Q[3][1]);
  b:=max(Q[1][2],Q[2][2],Q[3][2]);
  c:=max(Q[1][3],Q[2][3],Q[3][3]); d:=max(a,b,c); a2:=d-a;
  b2:=d-b; c2:=d-c; S:=orbit(T);
> for i from 0 to a2 do
  for j from 0 to b2 do
  for k from 0 to c2 do
    S:=S union orbit(addvect(T,[i,j,k]));
  > od; od; od; S; end;
```

Here the procedure *addvect* and *orbit* are:

```
1: addvect:=proc(T,v) local Q,a,b,c;
2: Q:=convert(T,list); a:=v[1]; b:=v[2]; c:=v[3];
3: {[Q[1][1]+a,Q[1][2]+b,Q[1][3]+c],[Q[2][1]+a,Q[2][2]+b,
  Q[2][3]+c],[Q[3][1]+a,Q[3][2]+b,Q[3][3]+c]};
4: end;
```

and

```
> orbit:=proc(T)local S,Q,T1;
> Q:=convert(T,list);
> T1:=[[Q[1][3],Q[1][2],Q[1][1]],[Q[2][3],Q[2][2],Q[2][1]],
  [Q[3][3],Q[3][2],Q[3][1]]];
> S:=orbit1(T) union orbit1(T1); S;
> end;
```

The *orbit1* takes care of the cube symmetries:

```
> orbit1:=proc(T) local
  i,k,T1,a,b,c,x,T2,T3,T4,T5,T6,T7,T8,T9,T10,T11,T12,T13,T14,
  T15,T16,T17,T18,T19,T20,T21,T22,T23,T24,S,Q,d,a1,b1,c1;
> Q:=convert(T,list);
> a:=max(Q[1][1],Q[2][1],Q[3][1]);
  a1:=min(Q[1][1],Q[2][1],Q[3][1]);
> b:=max(Q[1][2],Q[2][2],Q[3][2]);
  b1:=min(Q[1][2],Q[2][2],Q[3][2]);
> c:=max(Q[1][3],Q[2][3],Q[3][3]);
  c1:=min(Q[1][3],Q[2][3],Q[3][3]);
> d:=max(a,b,c); T1:=T;
> T2:=[[Q[1][2],Q[1][3],Q[1][1]],[Q[2][2],Q[2][3],Q[2][1]],
  [Q[3][2],Q[3][3],Q[3][1]]];
> T3:=[[Q[1][1],Q[1][3],Q[1][2]],[Q[2][1],Q[2][3],Q[2][2]],
  [Q[3][1],Q[3][3],Q[3][2]]];
> T4:=[[Q[1][1],Q[1][2],d-Q[1][3]],[Q[2][1],Q[2][2],d-Q[2][3]],
  [Q[3][1],Q[3][2],d-Q[3][3]]];
> T5:=[[Q[1][2],Q[1][3],d-Q[1][1]],[Q[2][2],Q[2][3],d-Q[2][1]],
  [Q[3][2],Q[3][3],d-Q[3][1]]];
```

```

> T6:={ [Q[1][1], Q[1][3], d-Q[1][2]], [Q[2][1], Q[2][3], d-Q[2][2]],
        [Q[3][1], Q[3][3], d-Q[3][2]] };
> T7:={ [Q[1][1], d-Q[1][2], Q[1][3]], [Q[2][1], d-Q[2][2], Q[2][3]],
        [Q[3][1], d-Q[3][2], Q[3][3]] };
> T8:={ [Q[1][2], d-Q[1][3], Q[1][1]], [Q[2][2], d-Q[2][3], Q[2][1]],
        [Q[3][2], d-Q[3][3], Q[3][1]] };
> T9:={ [Q[1][1], d-Q[1][3], Q[1][2]], [Q[2][1], d-Q[2][3], Q[2][2]],
        [Q[3][1], d-Q[3][3], Q[3][2]] };
> T10:={ [d-Q[1][1], Q[1][2], Q[1][3]], [d-Q[2][1], Q[2][2], Q[2][3]],
         [d-Q[3][1], Q[3][2], Q[3][3]] };
> T11:={ [d-Q[1][2], Q[1][3], Q[1][1]], [d-Q[2][2], Q[2][3], Q[2][1]],
         [d-Q[3][2], Q[3][3], Q[3][1]] };
> T12:={ [d-Q[1][1], Q[1][3], Q[1][2]], [d-Q[2][1], Q[2][3], Q[2][2]],
         [d-Q[3][1], Q[3][3], Q[3][2]] };
> T13:={ [Q[1][1], d-Q[1][2], d-Q[1][3]], [Q[2][1], d-Q[2][2],
        d-Q[2][3]], [Q[3][1], d-Q[3][2], d-Q[3][3]] };
> T14:={ [Q[1][2], d-Q[1][3], d-Q[1][1]], [Q[2][2], d-Q[2][3],
        d-Q[2][1]], [Q[3][2], d-Q[3][3], d-Q[3][1]] };
> T15:={ [Q[1][1], d-Q[1][3], d-Q[1][2]], [Q[2][1], d-Q[2][3],
        d-Q[2][2]], [Q[3][1], d-Q[3][3], d-Q[3][2]] };
> T16:={ [d-Q[1][1], d-Q[1][2], Q[1][3]], [d-Q[2][1], d-Q[2][2],
        Q[2][3]], [d-Q[3][1], d-Q[3][2], Q[3][3]] };
> T17:={ [d-Q[1][2], d-Q[1][3], Q[1][1]], [d-Q[2][2], d-Q[2][3],
        Q[2][1]], [d-Q[3][2], d-Q[3][3], Q[3][1]] };
> T18:={ [d-Q[1][1], d-Q[1][3], Q[1][2]], [d-Q[2][1], d-Q[2][3],
        Q[2][2]], [d-Q[3][1], d-Q[3][3], Q[3][2]] };
> T19:={ [d-Q[1][1], Q[1][2], d-Q[1][3]], [d-Q[2][1], Q[2][2],
        d-Q[2][3]], [d-Q[3][1], Q[3][2], d-Q[3][3]] };
> T20:={ [d-Q[1][2], Q[1][3], d-Q[1][1]], [d-Q[2][2], Q[2][3],
        d-Q[2][1]], [d-Q[3][2], Q[3][3], d-Q[3][1]] };
> T21:={ [d-Q[1][1], Q[1][3], d-Q[1][2]], [d-Q[2][1], Q[2][3],
        d-Q[2][2]], [d-Q[3][1], Q[3][3], d-Q[3][2]] };
> T22:={ [d-Q[1][1], d-Q[1][2], d-Q[1][3]], [d-Q[2][1], d-Q[2][2],
        d-Q[2][3]], [d-Q[3][1], d-Q[3][2], d-Q[3][3]] };
> T23:={ [d-Q[1][2], d-Q[1][3], d-Q[1][1]], [d-Q[2][2], d-Q[2][3],
        d-Q[2][1]], [d-Q[3][2], d-Q[3][3], d-Q[3][1]] };
> T24:={ [d-Q[1][1], d-Q[1][3], d-Q[1][2]], [d-Q[2][1], d-Q[2][3],
        d-Q[2][2]], [d-Q[3][1], d-Q[3][3], d-Q[3][2]] };
> S:={T1, T2, T3, T4, T5, T6, T7, T8, T9, T10, T11, T12, T13, T14, T15,
        T16, T17, T18, T19, T20, \break T21, T22, T23, T24};
> S; end:

```

Finally, we are ready to calculate the parameters in Theorem 2.1. We have $\alpha(T) = \text{transl}(T)$, $\beta(T) = \text{inters}(T)$ where

```

> inters:=proc(T) local a,b,c,Q,d,S,m,i,S1,S2;
        Q:=convert(T,list);
> a:=max(Q[1][1], Q[2][1], Q[3][1]);
> b:=max(Q[1][2], Q[2][2], Q[3][2]);
> c:=max(Q[1][3], Q[2][3], Q[3][3]);

```

```

> d:=max(a,b,c); S2:=transl(T); S:=convert(S2,list);
  m:=nops(S); S1:={};
> for i from 1 to m do S1:=S1 union {addvect(S[i],[0,0,1])};
  od;
> S2 intersect S1; end:

```

and $\gamma(T) = intersch(T)$ where

```

> intersch:=proc(T) local a,b,c,Q,d,S,m,i,S1,S2,S3,S4;
> Q:=convert(T,list);
> S2:=transl(T); S:=convert(S2,list); m:=nops(S); S1:={};
> for i from 1 to m do
  S1:=S1 union {addvect(S[i],[0,0,1])}; od; S3:={};
> for i from 1 m do S3:=S3 union {addvect(S[i],[0,1,0])}; od;
> nops(S1 intersect S3); end:

```

The Theorem 2.1 is then implemented in

$$f := (n, d, \alpha, \beta, \gamma) \rightarrow (n-d+1)^3\alpha - 3(n-d+1)^2(n-d)\beta + 3(n-d+1)(n-d)^2\gamma$$

```

> notrincn:=proc(T,n) local Q,a,b,c,x,a2,b2,c2,d,y,z,w;
> Q:=convert(T,list);
> a2:=max(Q[1][1],Q[2][1],Q[3][1]);
> b2:=max(Q[1][2],Q[2][2],Q[3][2]);
> c2:=max(Q[1][3],Q[2][3],Q[3][3]);
> d:=max(a2,b2,c2);
> x:=nops(transl(T)); y:=nops(inters(T)); w:=intersch(T);
> z:=f(n,d,x,y,w); end:

```

In the end one has to put all these procedures together and add the number of triangles together.

```

> main:=proc(p,lastside,nuptols)
> local i,j,k,s,nos,div,nod,nop,sol,x,netr,noft,l,z;
  netr:=nuptols;
> s:=sides(p); nos:=nops(s); print(s);
> for i from lastside to nos do
  div:=dkl(s[i]); nod:=nops(div);
> for j from 1 to nod do
  sol:=abcsol(div[j]); nop:=nops(sol);
> for k from 1 to nop do
> x:=minimaltr(s[i],sol[k][1],sol[k][2],sol[k][3],p);
> noft:=nops(x); if $noft>=1$ then
  for l from 1 to noft do
  z:=notrincn(x[l],p);
> netr:=netr+z; print(s[i],div,sol[k],x[l],checkeq(x[l]),
  z,netr,i);
> od; fi; od; od; od; netr; end:

```

The values $ET(n)$ for $n = 1 \dots 55$ computed with *main* are given next:

8	80	368	1264	3448	7792
16176	30696	54216	90104	143576	220328
326680	471232	664648	916344	1241856	1655208
2172584	2812664	3598664	4553800	5702776	7075264
8705088	10628928	12880056	15496616	18523472	22003808
26000584	30567400	35756776	41631672	48278136	55753272
64134536	73495760	83924408	95513248	108379264	122607856
138315720	155613408	174622488	195478424	218279240	243169352
270288064	299786440	331832248	366610560	404253120	444911712
488902856					

4. SOME FACTS AND CONJECTURES

If we look at the sequence $a_n = \frac{\ln ET(n)}{\ln(n+1)}$, $n \in \mathbb{N}$ it seems to be increasing. This sequence is clearly bounded from above since the number of all triangles in $\{0, \dots, n\}^3$ is not more than $(n+1)^9$ and so $a_n \leq 9$. Numerically, the best upper-bound for a_n seems to be a value slightly bigger than 5.


From what we have seen before, each class of triangles determined by a, b, c, d , solutions of (1), brings as a contribution i.e., a polynomial in terms of n . If we add these polynomials together, we get a polynomial which can be expressed of the variable $\zeta = n - 1$ ($n = \zeta + 1$) as follows:

$$\begin{aligned}
 n = 1: & \quad p_1(\zeta) = 8\zeta^3 + 24\zeta^2 + 24\zeta + 8, \\
 & \quad ET(1) = p_1(0) = 8; \\
 n = 2: & \quad p_2(\zeta) = p_1(\zeta) + 16 + 48(\zeta - 1) + 16(\zeta - 1)^3 + 48(\zeta - 1)^2, \\
 & \quad ET(2) = p_2(1) = 80; \\
 n = 3: & \quad p_3(\zeta) = p_2(\zeta) + 24 + 72(\zeta - 2) + 24(\zeta - 2)^3 + 72(\zeta - 2)^2, \\
 & \quad ET(3) = p_3(2) = 368; \\
 n = 4: & \quad p_4(\zeta) = p_3(\zeta) + 128 + 312(\zeta - 3) + 56(\zeta - 3)^3 + 240(\zeta - 3)^2, \\
 & \quad ET(4) = p_4(3) = 1264; \\
 n = 5: & \quad p_5(\zeta) = p_4(\zeta) + 40 + 120(\zeta - 4) + 40(\zeta - 4)^3 + 120(\zeta - 4)^2, \\
 & \quad ET(5) = p_5(4) = 3448; \\
 n = 6: & \quad p_6(\zeta) = p_5(\zeta) + 48 + 144(\zeta - 5) + 48(\zeta - 5)^3 + 144(\zeta - 5)^2, \\
 & \quad ET(6) = p_6(5) = 7792; \\
 n = 7: & \quad p_7(\zeta) = p_6(\zeta) + 776 + 1392(\zeta - 6) + 128(\zeta - 6)^3 + 744(\zeta - 6)^2, \\
 & \quad ET(7) = p_7(6) = 16176; \\
 n = 8: & \quad p_8(\zeta) = p_7(\zeta) + 232 + 552(\zeta - 7) + 88(\zeta - 7)^3 + 408(\zeta - 7)^2, \\
 & \quad ET(8) = p_8(7) = 30696; \\
 n = 9: & \quad p_9(\zeta) = p_8(\zeta) + 360 + 840(\zeta - 8) + 120(\zeta - 8)^3 + 600(\zeta - 8)^2, \\
 & \quad ET(9) = p_9(8) = 54216; \\
 n = 10: & \quad p_{10}(\zeta) = p_9(\zeta) + 80 + 80(\zeta - 9)^3 + 240(\zeta - 9)^2 + 240(\zeta - 9), \\
 & \quad ET(10) = p_{10}(9) = 90104; \\
 & \quad \dots
 \end{aligned}$$

We conjecture that in general

$$(7) \quad \begin{aligned} p_n(\zeta) = p_{n-1}(\zeta) + u_n(\zeta - n + 1)^3 + v_n(\zeta - n + 1)^2 \\ + w_n(\zeta - n + 1) + s_n, \end{aligned} \quad n \in \mathbb{N},$$

with u_n, v_n, w_n , and s_n non-negative integers.



As the graph above of $n \rightarrow \frac{\ln ET(n)}{\ln(n+1)}$ suggests, the second conjecture is that the following limit exists

$$(8) \quad \lim_{n \rightarrow \infty} \frac{\ln ET(n)}{\ln(n+1)} = C.$$

Acknowledgment. I would like to thank Ray Chandler who helped me solve the problem with an earlier version of the code which was giving some unwanted errors. Also I am thankful to Professor Rodrigo Obando who has translated the Maple code to Mathematica and calculated the first hundred terms.

REFERENCES

1. Ankeny N. C., *Sums of Three Squares*, Proceedings of AMS, **8**(2), 316–319.
2. Guy R., *Unsolved Problems in Number Theory*, Springer-Verlag, 2004.
3. Ionascu E. J., *A parametrization of equilateral triangles having integer coordinates*, Journal of Integer Sequences, **10** (2007), 09.6.7.
4. Rosen K., *Elementary Number Theory*, Fifth Edition, Addison Wesley, 2004

E. J. Ionascu, Department of Mathematics, Columbus State University, 4225 University Avenue, Columbus, GA 31907, USA, *e-mail*: ionascu_eugen@colstate.edu;